Práctica Cuarta
MATRICES Y APLICACIONES LINEALES

1.-Aplicacione lineales y matrices

1.1.-Toda aplicación lineal se representa por una matriz y toda matriz representa a una aplicación lineal.

Una misma aplicación lineal puede tener expresiones matriciales distintas en función de las bases elegidas

Vamos a ver un ejemplo de una aplicación lineal $f : R^3 \to R^4$ definida de la siguiente manera

In[1]:=

```
f[{x_, y_, z_}] := {x - y, x - z, 0, x}
```

¿Pero es lineal? Para que esto ocurra tiene que suceder que la imagen de una combinación lineal de vectores, sea la combinación lineal de las imágenes

In[2]:=

```
u = {u1, u2, u3};
v = {v1, v2, v3};
uno = a f[u] + b f[v];
dos = f[a u + b v];
Simplify[uno - dos]
```

Out[6]=

```
{0, 0, 0, 0}
```

Luego los vectores uno y dos son iguales

Consideremos las bases canónicas en los respectivos espacios

In[7]:=

```
A1 = IdentityMatrix[3];
 A2 = {{1, 0, 0, 0}, {0, 1, 0, 0}, {0, 0, 1, 0},
    {0, 0, 0, 1}};
```

¿Cuál es la matriz que caracteriza a esta aplicación lineal?

In[9]:=

```
matrizf = {};
For[i = 1, i ≤ 3, i++, AppendTo[matrizf, f[A1[[i]]]]];
matrizf = Transpose[matrizf];
MatrixForm[matrizf]
```

Out[12]//MatrixForm=

$$\begin{pmatrix} 1 & -1 & 0 \\ 1 & 0 & -1 \\ 0 & 0 & 0 \\ 1 & 0 & 0 \end{pmatrix}$$

De modo que la imagen de cualquier vector u , será

In[13]:=

```
matrizf.u
```

Out[13]=

```
{u1 - u2, u1 - u3, 0, u1}
```

In[14]:=

```
matrizf.{1, 2, 3}
```

Out[14]=

```
{-1, -2, 0, 1}
```

1.2 - Pero qué ocurrirá si las bases no son las canónicas ?

Sea g una aplicación lineal tal que $g : R^2 \to R^3$

In[15]:=

```
g[{x_, y_}] := {x + y, x - y, x}
```

Y sean

In[16]:=

```
B1 = {{1, 0}, {1, 1}};
B2 = {{1, 0, 1}, {0, 1, 1}, {1, 1, 0}};
```

bases de los espacios.

¿Cuál será la expresion matricial de la aplicación lineal en función de estas bases?

matriz de las imágenes de B1 por la aplicación g

In[18]:=

```
matriz1 = {g[B1[[1]]], g[B1[[2]]]}
```

Out[18]=

```
{{1, 1, 1}, {2, 0, 1}}
```

In[19]:=

```
B2
```

Out[19]=

```
{{1, 0, 1}, {0, 1, 1}, {1, 1, 0}}
```

expresion matricial es

In[20]:=

```
matriz = Inverse[B2].Transpose[matriz1]
```

```
General::spell1 :
 Possible spelling error: new symbol name "matriz" is
   similar to existing symbol "matrizf". More…
```

Out[20]=

$$\left\{\left\{\frac{1}{2}, \frac{3}{2}\right\}, \left\{\frac{1}{2}, -\frac{1}{2}\right\}, \left\{\frac{1}{2}, \frac{1}{2}\right\}\right\}$$

In[21]:=

```
MatrixForm[matriz]
```

Out[21]//MatrixForm=

$$\begin{pmatrix} \frac{1}{2} & \frac{3}{2} \\ \frac{1}{2} & -\frac{1}{2} \\ \frac{1}{2} & \frac{1}{2} \end{pmatrix}$$

Transformado de un vector según g

In[22]:=

```
{v1, v2}.Transpose[matriz]
```

Out[22]=

$$\left\{\frac{v1}{2}+\frac{3\,v2}{2},\ \frac{v1}{2}-\frac{v2}{2},\ \frac{v1}{2}+\frac{v2}{2}\right\}$$

In[23]:=

```
{1, 0}.Transpose[matriz]
```

Out[23]=

$$\left\{\frac{1}{2},\ \frac{1}{2},\ \frac{1}{2}\right\}$$

In[24]:=

```
{1, 1}.Transpose[matriz]
```

Out[24]=

$$\{2,\ 0,\ 1\}$$

Lo cual es correcto.

2.-Núcleo e imagen de una aplicación lineal

Sea g la aplicación anterior, cuya matriz es

In[25]:=

```
MatrixForm[matriz]
```

Out[25]//MatrixForm=

$$\begin{pmatrix} \frac{1}{2} & \frac{3}{2} \\ \frac{1}{2} & -\frac{1}{2} \\ \frac{1}{2} & \frac{1}{2} \end{pmatrix}$$

In[26]:=

```
Nucleodeg = NullSpace[matriz]
```

Out[26]=

```
{}
```

¿Que quiere decir? Que el núcleo de la aplicación lineal es el vector nulo

¿Quién es la imagen de una aplicación lineal?

Un sistema de generadores de la imagen de una aplicación lineal es la matriz formada por las columnas de la matriz que la representa

In[27]:=

```
generadorImagen = Transpose[matriz]
```

Out[27]=

$$\left\{\left\{\frac{1}{2}, \frac{1}{2}, \frac{1}{2}\right\}, \left\{\frac{3}{2}, -\frac{1}{2}, \frac{1}{2}\right\}\right\}$$

In[28]:=

```
RowReduce[generadorImagen]
```

Out[28]=

$$\left\{\left\{1, 0, \frac{1}{2}\right\}, \left\{0, 1, \frac{1}{2}\right\}\right\}$$

Luego estos dos vectores son una base de Im (f)

Sus ecuaciones son las siguientes

In[29]:=

```
parametros = {a, b};
coordenadas = {x, y, z};
parametricasImagen =
 LogicalExpand[
  coordenadas == Transpose[generadorImagen].parametros]
implicitasImagen =
 Eliminate[parametricasImagen, parametros]
```

Out[31]=

$$x == \frac{a}{2} + \frac{3\,b}{2}\ \&\&\ y == \frac{a}{2} - \frac{b}{2}\ \&\&\ z == \frac{a}{2} + \frac{b}{2}$$

Out[32]=

```
-y + 2 z == x
```

In[33]:=

```
Dimensions[Nucleodeg]
```

Out[33]=

```
{0}
```

In[34]:=

```
Dimensions[generadorImagen]
```

Out[34]=

```
{2, 3}
```

3.-Imagen de un subespacio vectorial

Sea una aplicación lineal entre j : $R^5 \to R^3$

Tal que su matriz sea

In[35]:=

```
matrizj = {{1, 0, 0, 1, 0}, {1, -1, 0, 0, 1},
  {1, -1, 0, 0, 0}};
```

```
General::spell :
 Possible spelling error: new symbol name "matrizj" is
   similar to existing symbols {matriz, matrizf}. More…
```

Queremos hallar la imagen del subespacio U dado por las siguientes ecuaciones implícitas

In[36]:=

```
implicitasU = {x + y - z == 0, y - z == 0, y + t == 0};
```

Para ello hacemos lo siguiente

In[37]:=

```
coordU = {x, y, z, t, w};
parametricasU = Solve[implicitasU, coordU]
```

```
Solve::svars : Equations may not give
   solutions for all "solve" variables. More…
```

Out[38]=

```
{{x → 0, y → -t, z → -t}}
```

In[39]:=

```
vectorU = coordU /. parametricasU[[1]]
```

Out[39]=

```
{0, -t, -t, t, w}
```

In[40]:=

```
parametros = {t, w};
baseU =
 Table[
  Table[Coefficient[vectorU[[j]], parametros[[i]]],
   {j, Length[vectorU]}], {i, Length[parametros]}]
```

Out[41]=

```
{{0, -1, -1, 1, 0}, {0, 0, 0, 0, 1}}
```

El sistema generador del subespacio imagen se halla calculando la imagen de cada uno de los vectores anteriores

In[42]:=

```
generadorV = Table[matrizj.baseU[[i]], {i, 2}]
```

Out[42]=

```
{{1, 1, 1}, {0, 1, 0}}
```

Y la ecuaciones son

In[43]:=

```
coordV = {x, y, z};
paramV = {a, b};
parametricasV =
 LogicalExpand[coordV == Transpose[generadorV].paramV]
implicitasV = Eliminate[parametricasV, paramV]
```

```
General::spell1 :
 Possible spelling error: new symbol name "coordV" is
   similar to existing symbol "coordU". More…
```

```
General::spell1 : Possible spelling error:
   new symbol name "parametricasV" is similar
   to existing symbol "parametricasU". More…
```

Out[45]=

$x == a \,\&\&\, y == a + b \,\&\&\, z == a$

```
General::spell1 : Possible spelling error:
   new symbol name "implicitasV" is similar
   to existing symbol "implicitasU". More…
```

Out[46]=

$z == x$

Created by Mathematica (January 28, 2004)